
README
    

  
  


  Telemetry-based metrics UI. Take your telemetry metrics and display them in a web page.
  


  
Features
TelemetryUI’s primary goal is to display your application metrics without external infrastructure dependencies. Plug, Phoenix, Phoenix LiveView, Absinthe, Ecto, Erlang VM, Tesla, Finch, Redix, Oban, Broadway and others expose all sorts of data that can be useful. You can also emit your own events from your application.
Your data should not have to be uploaded somewhere else to have insighful metrics.
It comes with a Postgres backend, powered by Ecto, to quickly (and efficiently) store and query your application events.
[image: Screenshot of /metrics showcasing values and charts]Advantages over other tools
	Persisted metrics inside your own database
	Live dashboard
	Many built-in charts and visualizations using VegaLite

Advanced features
	100% custom UI hook to show your own components
	100% custom data fetching to show live data
	Shareable metrics page (secured, cacheable, without external requests)
	Slack digest with rendered images
	Multiple metrics dashboard living in the same app

Checkout the Guides for more informations.
Usage
Installation
TelemetryUI is published on Hex. Add it to your list of dependencies in mix.exs:
# mix.exs
def deps do
  [
    {:telemetry_ui, "~> 4.0"}
  ]
end
Configure TelemetryUI for test.
# config/test.exs
config :telemetry_ui, disabled: true
Then run mix deps.get to install Telemetry and its dependencies.
After the packages are installed you must create a database migration to add the telemetry_ui_events table to your database:
mix ecto.gen.migration add_telemetry_ui_events_table

Open the generated migration in your editor and call the up and down functions on TelemetryUI.Adapter.EctoPostgres.Migrations:
defmodule MyApp.Repo.Migrations.AddTelemetryUIEventsTable do
  use Ecto.Migration

  @disable_migration_lock true
  @disable_ddl_transaction true

  def up do
    TelemetryUI.Backend.EctoPostgres.Migrations.up()
  end

  # We specify `version: 1` in `down`, ensuring that we'll roll all the way back down if
  # necessary, regardless of which version we've migrated `up` to.
  def down do
    TelemetryUI.Backend.EctoPostgres.Migrations.down(version: 1)
  end
end
This will run all of TelemetryUI's versioned migrations for your database. Migrations between versions are idempotent and rarely change after a release. As new versions are released you may need to run additional migrations.
Now, run the migration to create the table:
mix ecto.migrate

TelemetryUI instances are isolated supervision trees and must be included in your application's supervisor to run. Use the application configuration you've just set and include TelemetryUI in the list of supervised children:
# lib/my_app/application.ex
def start(_type, _args) do
  children = [
    MyApp.Repo,
    {TelemetryUI, telemetry_config()}
  ]

  Supervisor.start_link(children, strategy: :one_for_one, name: MyApp.Supervisor)
end

defp telemetry_config do
  import TelemetryUI.Metrics

  [
    metrics: [
      last_value("my_app.users.total_count", description: "Number of users", ui_options: [unit: " users"]),
      counter("phoenix.router_dispatch.stop.duration", description: "Number of requests", unit: {:native, :millisecond}, ui_options: [unit: " requests"]),
      value_over_time("vm.memory.total", unit: {:byte, :megabyte}),
      distribution("phoenix.router_dispatch.stop.duration", description: "Requests duration", unit: {:native, :millisecond}, reporter_options: [buckets: [0, 100, 500, 2000]]),
    ],
    backend: %TelemetryUI.Backend.EctoPostgres{
      repo: MyApp.Repo,
      pruner_threshold: [months: -1],
      pruner_interval_ms: 84_000,
      max_buffer_size: 10_000,
      flush_interval_ms: 10_000
    }
  ]
end
Since the config is read once at startup, you need to restart the server of you add new metrics to track.
To see the rendered metrics, you need to add a route to your router.
# lib/my_app_web/router.ex
scope "/" do
  get("/metrics", TelemetryUI.Web, [], [assigns: %{telemetry_ui_allowed: true}])
end
Security
Since it may contain sensitive data, TelemetryUI requires a special assign to render the page.
:telemetry_ui_allowed must be set to true in the conn struct before it enters the TelemetryUI.Web module.
By using a special assign to control access, you can integrate TelemetryUI.Web page with your existing authorization. We can imagine an admin protected section that also gives you access to the TelemetryUI.Web page:
pipeline :admin_protected do
  plug(MyAppWeb.EnsureCurrentUser)
  plug(MyAppWeb.EnsureRole, :admin)
  plug(:enable_telemetry_ui)
end

def enable_telemetry_ui(conn, _), do: assign(conn, :telemetry_ui_allowed, true)
That’s it! You can declare as many metrics as you want and they will render in HTML on your page!
License
TelemetryUI is © 2023 Mirego and may be freely distributed under the New BSD license. See the LICENSE.md file.
About Mirego
Mirego is a team of passionate people who believe that work is a place where you can innovate and have fun. We’re a team of talented people who imagine and build beautiful Web and mobile applications. We come together to share ideas and change the world.
We also love open-source software and we try to give back to the community as much as we can.



  

    
Using application data
    

Sometimes we already have metrics that we want to display without going through the telemetry package.
We want to expose the number of users created. The users table with the inserted_at is perfect for this.
telemetry_ui provides a way use the existing metrics helpers with application-defined data:
count_over_time(:data,
  description: "Users count",
  unit: " users",
  data_resolver: fn options ->
    query =
      from(
        users in "users",
        select: %{date: users.inserted_at, count: 1},
        where: users.inserted_at >= ^options.from and users.inserted_at <= ^options.to
      )

    {:ok, MyApp.Repo.all(query)}
  end
)
The data_resolver options is specified here to return the data that will be displayed in the same graph layout as the telemetry_ui_events entries.
The built-in metrics uses the same data_resolver options: data_resolver: &{:async, fn -> TelemetryUI.metric_data(&1, metric, &2) end}.



  

    
Custom UI
    

telemetry_ui offers many way to customize the layouts, colors and components of your dashboard.
Theme
The theme option of you configuration can be customized:
[
  metrics: [
    {"Users", [count_over_time("myapp.users.created")]}
  ],
  backend: backend(),
  theme: %{
    header_color: "blue",
    title: "User dashboard metrics"
  }
]
Here is the list of available options:
	header_color: A CSS color used as the text color in the header
	title: The title used in the header
	logo: An SVG binary used in the header and the favicon
	scale: Available colors used in graph. The colors are used in the same order they are defined in.
	share_key: 16 characters key that enable the sharing of a dashboard page. The share feature is hidden when the option is nil
	share_path: Application router path to render the share pages.
	frame_options: List of options for the time frame select in the UI. Format: {atom_identitifer, number, unit} {:last_3_minutes, 3, :minute}

Layout
The list of metrics can also be customized with TailwindCSS classes. Here is an example with 2 metrics side-by-side in a grid:
metrics = [
  distribution("my_app.repo.query.total_time",
    reporter_options: [buckets: [0, 4, 10, 50]],
    ui_options: [class: "col-span-2"],
    unit: {:native, :millisecond}
  ),
  distribution("phoenix.router_dispatch.stop.duration",
    reporter_options: [buckets: [0, 40, 100, 500]],
    ui_options: [class: "col-span-2"],
    unit: {:native, :millisecond}
  ),
  distribution("phoenix.router_dispatch.stop.duration",
    tags: [:route],
    reporter_options: [buckets: [0, 40, 100, 500]],
    ui_options: [class: "col-span-full"],
    unit: {:native, :millisecond}
  )
]

[
  metrics: [
    {"Metrics layout page", metrics, ui_options: [metrics_class: "grid-cols-4 gap-4"]}
  ],
  backend: #...
  theme: #...
]
All grid classes are available with sm and lg variant to support mobile and desktop viewports.



  

    
Multiple metrics endpoints
    

The /metrics endpoint exposes different pages of metrics. We can also have multiple metrics endpoint.
The benefit is to have different metrics exposed with different permissions. Here is an example:
We have our router defining routes with different telemetry_ui_name Plug.Conn assign.
scope "/admin" do
  pipe_through [EnsureAdmin]

  get("/metrics", TelemetryUI.Web, :index, assigns: %{telemetry_ui_name: :admin})
end

scope "/user_dashboard" do
  pipe_through [EnsureUser]

  get("/metrics", TelemetryUI.Web, :index, assigns: %{telemetry_ui_name: :user_dashboard})
end
defmodule MyApp.Telemetry do
  import TelemetryUI.Metrics

  def admin do
    [
      name: :admin,
      metrics: [
        {"Memory", [value_over_time("vm.memory.total", unit: {:byte, :megabyte})]}
      ],
      backend: backend(),
      theme: %{header_color: "purple", title: "Admin metrics"}
    ]
  end

  def user_dashboard do
    [
      name: :user_dashboard,
      metrics: [
        {"Followers", [count_over_time("myapp.users.new_follower")]}
      ],
      backend: backend(),
      theme: %{header_color: "blue", title: "User dashboard metrics"}
    ]
  end

  defp backend do
    %TelemetryUI.Backend.EctoPostgres{
      repo: MyApp.Repo,
      pruner_threshold: [months: -1],
      pruner_interval_ms: 84_000,
      max_buffer_size: 1,
      flush_interval_ms: 1,
      verbose: true
    }
  end
end
Don’t forget to start all TelemetryUI process in application.ex:
defmodule MyApp.Application do
  use Application

  def start(_type, _args) do
    children = [
      # ...
      {TelemetryUI, MyApp.Telemetry.admin()},
      {TelemetryUI, MyApp.Telemetry.user_dashboard()}
    ]

    opts = [strategy: :one_for_one, name: MyApp.Supervisor]
    Supervisor.start_link(children, opts)
  end
end



  

    
TelemetryUI.Config 
    



      
Holds the state of the config.
See other modules for more details:
	TelemetryUI.Theme
	TelemetryUI.Backend

TelemetryUI 
    



      
Main entry point to start all the processes as part of your application.
Usage
use Application

def start(_type, _args) do
  children = [
    Shopcast.Repo,
    MyAppWeb.Endpoint,
    {TelemetryUI, my_config()}
  ]
The reporter, write buffer and pruner processes will be part of your supervision tree, ensuring that everything runs smoothly.
The config is a data structure representing all settings used by TelemetryUI.
Config example
  [
    metrics: [
      {"System", [last_value("vm.memory.total", unit: {:byte, :megabyte})]}
    ],
    theme: %{title: "Metrics"},
    backend: 
      %TelemetryUI.Backend.EctoPostgres{
        repo: MyApp.Repo,
        pruner_threshold: [months: -1],
        pruner_interval_ms: 3_600_000,
        max_buffer_size: 1_000,
        flush_interval_ms: 10_000
      }
  ]
This config will show a basic "last value" chart for the memory usage reported by the VM.
The title of the page will be "Metrics" and the backend used to store and query metrics will be PostgreSQL.
See TelemetryUI.Config for a list of all options.
Metrics list
Every kind of metrics exposed by TelemetryMetrics is supported by TelemetryUI:
	last_value
	summary
	sum
	counter
	distribution

TelemetryUI also exposes its own set of metrics:
	average
	average_over_time
	count_over_time
	median
	median_over_time

The Web
Finally, when the configuration is done and we actually want to see our metrics, we need to add the TelemetryUI.Web module in our router:
Phoenix
scope "/" do
  pipe_through([:browser])
  get("/metrics", TelemetryUI.Web, [], [assigns: %{telemetry_ui_allowed: true}])
end
The metrics page is protected by default. It needs to have the telemetry_ui_allowed assign to true to render.
We can imagine having a :admin_protected plug that ensure a user is an admin and also assign telemetry_ui_allowed to true.
scope "/" do
  pipe_through([:browser, :admin_protected])
  get("/metrics", TelemetryUI.Web, [])
end
That’s it! The /metrics page will show the metrics as they are recorded. Checkout the Guides to dive into more complex configuration and awesome features :)

TelemetryUI.Theme 
    



      
Customize the UI of the web view.
	primary_color: Example: #ff0000, Used for the title and first color on the charts
	title: Example: "My metrics", Used for the HTML page title and the title in the header.
	description: Example: "HTTP, GraphQL and Database metrics", Used for the HTML description tag.
	logo: SVG logo in string used for the favicon and header.
	scale: List of hex colors used in the charts.
	share_key: 16 characters string used to hash the shared page.
	share_path: Path in string starting by a / used to generate the shared page. It’s difference from the metrics page since it does not require authentication.
	frame_options: Example: [{:last_30_minutes, 30, :minute}, {:last_2_hours, 120, :minute}], Used to generate the time frame in the header.


      





  

    
TelemetryUI.Backend protocol
    



      
Inserting, fetching and pruning of metrics.
Inserting
Inserting is called by the internal WriteBuffer. It can be grouped depending on the buffer configuration inside the backend struct:
	flush_interval_ms: Time interval before the write buffer calls the backend
	max_buffer_size: Maximum count of events before the write  buffer calls the backend

Fetching
Fetching is called when rendering metrics in the view. The filters argument is a struct with predefined fields: to, from, event_name and compare.
Pruning
Pruning is implemented to keep the datastore clean. Keeping data forever will increase the size of the storage and affect performance.
	pruner_threshold: Example: [months: -1]. Delete events older than a month. 
	pruner_interval_ms: Example: 84_000. Time interval for the pruner process to run. The process simply calls #prune_events!/2.

TelemetryUI.Backend.EctoPostgres.Migrations 
    



      
Migrations create and modify the database tables TelemetryUI needs to function.
Usage
To use migrations in your application you'll need to generate an Ecto.Migration that wraps
calls to TelemetryUI.Backend.EctoPostgres.Migrations:
mix ecto.gen.migration add_telemetry_ui

Open the generated migration in your editor and call the up and down functions on
TelemetryUI.Backend.EctoPostgres.Migrations:
defmodule MyApp.Repo.Migrations.AddTelemetryUI do
  use Ecto.Migration

  def up, do: TelemetryUI.Backend.EctoPostgres.Migrations.up()

  def down, do: TelemetryUI.Backend.EctoPostgres.Migrations.down()
end
This will run all of TelemetryUI's versioned migrations for your database.
Now, run the migration to create the table:
mix ecto.migrate

Migrations between versions are idempotent. As new versions are released, you
may need to run additional migrations. To do this, generate a new migration:
mix ecto.gen.migration upgrade_telemetry_ui_to_v11

Open the generated migration in your editor and call the up and down
functions on TelemetryUI.Backend.EctoPostgres.Migrations, passing a version number:
defmodule MyApp.Repo.Migrations.UpgradeTelemetryUIToV11 do
  use Ecto.Migration

  def up, do: TelemetryUI.Backend.EctoPostgres.Migrations.up(version: 11)

  def down, do: TelemetryUI.Backend.EctoPostgres.Migrations.down(version: 11)
end
Isolation with Prefixes
TelemetryUI supports namespacing through PostgreSQL schemas, also called "prefixes" in Ecto. With
prefixes your events table can reside outside of your primary schema (usually public) and you can
have multiple separate job tables.
To use a prefix you first have to specify it within your migration:
defmodule MyApp.Repo.Migrations.AddPrefixedTelemetryUIJobsTable do
  use Ecto.Migration

  def up, do: TelemetryUI.Backend.EctoPostgres.Migrations.up(prefix: "private")

  def down, do: TelemetryUI.Backend.EctoPostgres.Migrations.down(prefix: "private")
end
The migration will create the "private" schema and all tables, functions and triggers within
that schema. With the database migrated you'll then specify the prefix in your configuration:
config :my_app, TelemetryUI,
  prefix: "private",
  ...
In some cases, for example if your "private" schema already exists and your database user in
production doesn't have permissions to create a new schema, trying to create the schema from the
migration will result in an error. In such situations, it may be useful to inhibit the creation
of the "private" schema:
defmodule MyApp.Repo.Migrations.AddPrefixedTelemetryUIJobsTable do
  use Ecto.Migration

  def up, do: TelemetryUI.Backend.EctoPostgres.Migrations.up(prefix: "private", create_schema: false)

  def down, do: TelemetryUI.Backend.EctoPostgres.Migrations.down(prefix: "private")
end

Run all migrations from current version down to the first:
TelemetryUI.Backend.EctoPostgres.Migrations.down()
Run migrations down to and including a specified version:
TelemetryUI.Backend.EctoPostgres.Migrations.down(version: 5)
Run migrations in an alternate prefix:
TelemetryUI.Backend.EctoPostgres.Migrations.down(prefix: "payments")

Run the up changes for all migrations between the initial version and the current version.

Run all migrations up to the current version:
TelemetryUI.Backend.EctoPostgres.Migrations.up()
Run migrations up to a specified version:
TelemetryUI.Backend.EctoPostgres.Migrations.up(version: 2)
Run migrations in an alternate prefix:
TelemetryUI.Backend.EctoPostgres.Migrations.up(prefix: "payments")
Run migrations in an alternate prefix but don't try to create the schema:
TelemetryUI.Backend.EctoPostgres.Migrations.up(prefix: "payments", create_schema: false)

Plug to render an HTML view with all metrics.
The view handles the different pages in the configurationa and the assets pipeline for CSS and JavaScript.
The module also handles "async" components data request called in the components.

